
	  function f_MCX_InviteMT_SendINVITE(MCX_CallType_Type p_CallType := ANY,
                                     MCX_PriorityCallType_Type p_PriorityCallType := NORMAL_CALL,
                                     MCX_CommencementMode_Type p_CommencementMode := NONE,
                                     template (omit) Contact p_Contact := omit,
                                     template (omit) SDP_Message p_SDP_Message := omit,
                                     template (value) MCX_MIME_XmlBodies_Type p_XmlBodies := cs_MCX_MIME_XmlBodies_AllOmit,
                                     boolean p_IsReInvite := false)
    runs on MCX_IMS_PTC return IMS_InviteRequestWithSdp_Type
  { /* p_Contact: if omit then the default Contact header according to 36.579-1 Table 5.5.2.5.2-1 is used */
    var MCX_ServiceType_Type v_ServiceType := MCPTT;
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;
    var template (value) INVITE_Request v_InviteRequest;
    var template (value) MessageHeader v_MessageHeader_Invite;
    var template (omit) MessageBody v_MessageBody := omit;
    var template (omit) ContentType v_ContentType := omit;
    var SipUrl v_RequestUrl;

    f_MCX_CallSetup_ConfigSRTP(p_CallType, IPCAN_MT_SpeechCall);

    if (not p_IsReInvite) {
      v_RequestUrl := f_IMS_PTC_ImsInfo_GetContactUrl();         /* SIP URI of the UE's contact address as provided in the Contact-header of the REGISTER message */
      f_MCX_CallSetup_ConfigSRTP(p_CallType, IPCAN_MT_SpeechCall);
      f_MCX_IPCAN_StartSignalling(IPCAN_MT_SpeechCall);
    } else {
      v_RequestUrl := f_IMS_PTC_ImsInfo_GetContactUrl(dialog);   /* SIP URI of the UE's contact address as provided in the Contact-header within any response or request within the dialog */
    }

    if (isvalue(p_XmlBodies.McpttInfo) or isvalue(p_XmlBodies.McpttLocationInfo)) {
      v_MessageBody := f_MCX_MIME_MessageBodyTX(v_ServiceType, -, p_SDP_Message, p_XmlBodies.McpttInfo, p_XmlBodies.McpttLocationInfo);
      v_ContentType := cs_ContentTypeMIME(tsc_MCX_MIME_Boundary);
    } else if (isvalue(p_SDP_Message)) {
      v_MessageBody := cs_MessageBody_SDP(valueof(p_SDP_Message));
      v_ContentType := cs_ContentTypeSDP;
    }

    v_MessageHeader_Invite := f_MCX_InviteRequest_MessageHeaderTX(MCPTT, p_PriorityCallType, p_CommencementMode, p_Contact, v_ContentType, p_IsReInvite);
    v_InviteRequest := cs_INVITE_Request(v_RequestUrl, v_MessageHeader_Invite, v_MessageBody);
    v_InviteRequestWithSdp := f_IMS_InviteRequestWithSdpMT(v_InviteRequest, p_SDP_Message);
    IMS_Client.send(cas_IMS_Invite_Request(v_InviteRequestWithSdp.RoutingInfo, v_InviteRequest));

    return v_InviteRequestWithSdp;
  }

	  function f_MCX_InviteMO_ReceiveINVITE(MCX_ServiceType_Type p_ServiceType := MCPTT,
                                        MCX_CallType_Type p_CallType,
                                        MCX_PriorityCallType_Type p_PriorityCallType := NORMAL_CALL,
                                        MCX_CommencementMode_Type p_CommencementMode := AUTOMATIC,
                                        charstring p_CalleeUri := tsc_MCPTT_PublicServiceId_A,
                                        template (present) MCX_SdpInfoRX_Type p_SdpInfoRX,
                                        template (present) MCX_MIME_XmlBodies_Type p_XmlBodies,
                                        boolean p_IsReInvite := false)
    runs on MCX_IMS_PTC return IMS_InviteRequestWithSdp_Type
  {
    var template (present) SipUrl v_RequestUri := f_SIP_BuildSipUri_RX(p_CalleeUri);
    var template (present) MessageHeader v_MessageHeader_Invite := f_MCX_InviteRequest_MessageHeaderRX(p_ServiceType, -, p_PriorityCallType, p_CommencementMode, v_RequestUri, p_IsReInvite);
    var IMS_DATA_REQ v_IMS_DATA_REQ;
    var MIME_Message v_MIME_Message;
    var SDP_Message v_SDP_Message;
    var MCX_MIME_XmlBodies_Type v_XmlBodies;
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;
    var boolean v_AlertInd;
    
    v_IMS_DATA_REQ := valueof(f_MCX_InviteMO_ReceiveRequest_Optional(v_RequestUri, v_MessageHeader_Invite, cr_MessageBody_MIME, -, p_IsReInvite));
    v_MIME_Message := v_IMS_DATA_REQ.Request.Invite.messageBody.mimeMessageBody;
    v_SDP_Message := valueof(f_MIME_DecodeMatchAndCheckSDP(v_MIME_Message, p_SdpInfoRX));
    v_InviteRequestWithSdp := f_IMS_InviteRequestWithSdpMO(v_IMS_DATA_REQ, v_SDP_Message);

    f_MCX_CallSetup_ConfigSRTP(p_CallType, IPCAN_MT_SpeechCall);

    if (not p_IsReInvite) {
      f_MCX_CallSetup_ConfigSRTP(p_CallType, IPCAN_MO_SpeechCall);
      f_IMS_IPCAN_SendCoOrdMsg(IPCAN);       /* trigger IPCAN to continue with activation of dedicated EPS bearer */
    }
    v_XmlBodies := f_MIME_DecodeAndMatchXML(v_MIME_Message, v_ServiceType, -, p_XmlBodies.McpttInfo, -, -, p_XmlBodies.McpttLocationInfo, p_XmlBodies.McvideoLocationInfo, p_XmlBodies.McdataLocationInfo, p_XmlBodies.ResourceLists);
    if (p_PriorityCallType == EMERGENCY_CALL) {     // the UE may set alert-ind:=true in which case it shall include McpttLocationInfo
      v_AlertInd := isvalue(v_XmlBodies.McpttInfo.mcptt_Params.alert_ind) and match(v_XmlBodies.McpttInfo.mcptt_Params.alert_ind, cr_MCPTT_Content_True);
      if (v_AlertInd and not isvalue(v_XmlBodies.McpttLocationInfo)) {
        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "alert-ind==true in McpttInfo but McpttLocationInfo is not present");
      } else if (not v_AlertInd and isvalue(v_XmlBodies.McpttLocationInfo)) {
        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "alert-ind==false in McpttInfo but McpttLocationInfo is present");
      }
    }

    return fl_MCX_InviteRequestWithSdpAndXmlMO(v_IMS_DATA_REQ, v_SDP_Message, v_XmlBodies);
  }

